CE-580

COMPUTATIONAL TECHNIQUES

FOR

FLUID DYNAMICS

HOMEWORK #5

Numerical Solution to Water Waves

Taha Yaşar Demir

1881978

CE 580 COMPUTATIONAL TECHNIQUES FOR FLUID DYNAMICS

Homework 5

Water Surface Profile in a Rectangular Basin with Sinusoidal Waves on Far-End

Sinusoidal water waves are applied at the right end of a horizontal rectangular basin. The other end is a vertical wall. Write a computer program to obtain water surface profile for the wave motion in the channel.



1. Apply integral continuity and momentum equations for a control volume assuming one-dimensional flow along the channel.
2. Use two-step Lax-Wendroff scheme for the numerical solution of the governing equations.
3. Apply a sinusoidal wave at the right end of the channel as the boundary condition.
4. Generate a computational domain with 2001 nodes for constant mesh size.
5. Use a constant time step $\Delta t=0.0008$ s. Also, estimate the maximum allowable time step size for each computational step.
6. Run your program for 10 wave cycles for the data given below:

Channel width	$W = 1.$ m
Channel length	$L = 20.$ m
Initial water height	$h_0 = 12.$ m
Initial velocity	$V_0 = 0.$
Friction coefficient	$C_f = 0.005$
Wave period	$T = 2.$ s
Wave amplitude	$H = 1.$ m

7. Make a plot of the allowable time-step size as a function of time.
8. Make a plot of the water surface level at both ends of the basin as function of time (on the same graph).
9. Make a plot of the water surface profile in the basin at the end of 10th wave cycle.
10. Run your program for larger mesh sizes and compare the water surface profiles at the end of 10th wave cycle.
11. Discuss the results you obtained.

Computations

Consider the control volume shown in figure 1



Figure 1: Control Volume

Assumptions:

- Quasi-One-dimensional flow
- Velocity is uniform across vertically
- Incompressible fluid

Governing Equations

The integral continuity equation

$$\int_{cs} \rho V.dA + \frac{\partial}{\partial t}\int_{cv} \rho d\forall = 0$$

Applied to above control volume

$$-\rho u_i w h_i + \rho u_{i+1} w h_{i+1} + \frac{\partial}{\partial t}\int_{x}^{x+\Delta x} \rho h w dx = 0$$

Note that, since the dependent variables above is not dependent on x we can take time derivative inside

$$\frac{\partial}{\partial t}\int_{x}^{x+\Delta x} \rho h w dx = \int_{x}^{x+\Delta x} \frac{\partial}{\partial x}(\rho h w) dx$$

Regrouping the continuity and dividing by $\rho w \Delta x$

$$\frac{(uh)_{i+1} - (uh)_i}{\Delta x} + \frac{1}{\Delta x}\int_{x}^{x+\Delta x} \frac{\partial h}{\partial t} dx = 0$$

As Δx goes to 0, continuity equation becomes

$$\frac{\partial (uh)}{\partial x} + \frac{\partial h}{\partial t} = 0 \quad (1)$$

The integral momentum equation in x-direction is

$$F_{Sx} + F_{Bx} = \int_{cs} u(\rho V.dA) + \frac{\partial}{\partial t}\int_{cv} u(\rho d\forall)$$

Applying to the CV gives

$$\frac{\gamma h_i^2 w}{2} + \frac{\gamma h_{i+1}^2 w}{2} - F_f = \rho u_{i+1}^2 w h_{i+1} - \rho u_i^2 w h_i + \frac{\partial}{\partial t}\int_{x}^{x+\Delta x} \rho u w h dx$$

The friction force can be related to the shear stress, and it also depends on flow direction

$$F_f = \tau_w \Delta x p_w$$

Where τ_w is the shear stress and $p_w(= w + 2h)$ is the wetted perimeter.

The wall shear stress can be expressed in terms of a friction coefficient based on the local cross sectional average velocity

$$C_f = \frac{\tau_w}{\left(\frac{\rho u^2}{2}\right)} \quad then,$$

$$F_f = \frac{C_f \rho u|u|\Delta x p_w}{2}$$

Substituting into the momentum equation and dividing by $\rho w \Delta x$

$$\frac{1}{\Delta x}\left[\frac{g h_i^2}{2} - \frac{g h_{i+1}^2}{2}\right] - \frac{Cfu|u|p_w}{2w} = \frac{(hu^2)_{i+1} - (hu^2)_i}{\Delta x} + \frac{1}{\Delta x}\int_{x}^{x+\Delta x} \frac{\partial (uh)}{\partial t} dx$$

As Δx goes to 0 yields momentum equation

$$\frac{\partial (uh)}{\partial t} + \frac{\partial}{\partial x}\left[hu^2 + \frac{gh^2}{2}\right] = -\frac{C_f u|u|p_w}{2w} \quad (2)$$

The continuity equation (1) and momentum equation (2) represent the two equations to be solved to yield the water surface profile and the velocity distribution. For convenience we will introduce new variables for the parametric appearing in the equations

$$uh = A \ , \qquad hu^2 + \frac{gh^2}{2} = B \ , \qquad \frac{C_f u|u|p_w}{2w} = C$$

In terms of new variable, the equations simplify to

$$\frac{\partial h}{\partial t} + \frac{\partial A}{\partial x} = 0 \ , \qquad \frac{\partial A}{\partial t} + \frac{\partial B}{\partial x} = C$$

Grid System



Figure 2: Grid System

Discretization

Two step Lax-Wendrof Method is used for numerical solution of these equations

The continuity equation

$$\frac{\partial h}{\partial t} + \frac{\partial A}{\partial x} = 0$$

$$\frac{h_i^{i+\frac{1}{2}} - \frac{h_{i-1}^n + h_{i+1}^n}{2}}{\frac{\Delta t}{2}} + \frac{A_{i+1}^n - A_{i-1}^n}{2\Delta x} = 0$$

Solving for $h_i^{n+\frac{1}{2}}$

$$h_i^{n+\frac{1}{2}} = \frac{1}{2}(h_{i+1}^n + h_{i-1}^n) - \frac{\Delta t}{4\Delta x}(A_{i+1}^n - A_{i-1}^n)$$

The momentum equation

$$\frac{\partial A}{\partial t} + \frac{\partial B}{\partial x} = C$$

Discretized as

$$A_i^{n+1} = \frac{1}{2}(A_{i+1}^n + A_{i-1}^n) - \frac{\Delta t}{4\Delta x}(B_{i+1}^n - B_{i-1}^n) - \frac{\Delta t C_i^n}{2}$$

At completion of this step, one calculates velocity and other variables

$$u_i^{n+\frac{1}{2}} = \frac{A_i^{n+\frac{1}{2}}}{h_i^{n+\frac{1}{2}}}\ , \qquad B_i^{n+\frac{1}{2}} = \left(hu^2 + \frac{gh^2}{2}\right)_i^{n+\frac{1}{2}}\ , \qquad C_i^{n+\frac{1}{2}} = \left(\frac{C_f u|u|p_w}{2w}\right)_i^{n+\frac{1}{2}}$$

The second step proceeds from n to n+1 using the variables at the intermediate step to evaluate the spatial derivatives. This is known as a ‘leapfrog’ step and suggests a central difference formulation for derivatives

The second step for continuity

$$\frac{h_i^{n+1} - h_i^n}{\Delta t} + \frac{A_{i+1}^{n+\frac{1}{2}} - A_{i-1}^{n+\frac{1}{2}}}{2\Delta x} = 0 \quad or$$

$$h_i^{n+1} = h_i^n - \frac{\Delta t}{2\Delta x}\left(A_{i+1}^{n+\frac{1}{2}} - A_{i-1}^{n+\frac{1}{2}}\right)$$

And the second step for momentum gives

$$A_i^{n+1} = A_i^n - \frac{\Delta t}{2\Delta x}\left(B_{i+1}^{n+\frac{1}{2}} - B_{i-1}^{n+\frac{1}{2}}\right) - \Delta t C_i^{n+\frac{1}{2}}$$

And then

$$u_i^{n+1} = \frac{A_i^{n+1}}{h_i^{n+1}} \quad , \qquad B_i^{n+1} = \left(hu^2 + \frac{gh^2}{2}\right)_i^{n+1} \quad , \qquad C_i^{n+1} = \left(\frac{C_f u|u|p_w}{2w}\right)_i^{n+1}$$

Boundary Conditions

At two ends different discretization is necessary, since central difference on A and B is not possible

At the wall, i=1, forward differencing is used to obtain $h_i^{n+\frac{1}{2}}$ at intermediate time step

$$h_i^{n+\frac{1}{2}} = (h_1^n) - \frac{\Delta t}{2\Delta x}(A_2^n - A_1^n) \;, \qquad (u_1 = 0, A_1 = 0)$$

For second step to n+1i the values for h at the wall is

$$h_i^{n+1} = (h_1^n) - \frac{\Delta t}{\Delta x}\left(A_2^{n+\frac{1}{2}} - A_1^{n+\frac{1}{2}}\right) \;, \qquad (u_1 = 0, A_1^{n+\frac{1}{2}} = 0)$$

At the reservoir side,I=M, h is determined from input wave function

$$h_M = h_0 + \frac{H}{2}\sin\left(\frac{2\pi t}{T}\right)$$

And A is determined such as

$$A_M^{n+\frac{1}{2}} = A_M^n - \frac{\Delta t}{2\Delta x}(B_M^n - B_{M-1}^n) - \frac{\Delta t C_M^n}{2}$$

$$A_M^{n+1} = A_M^n - \frac{\Delta t}{\Delta x}\left(B_M^{n+\frac{1}{2}} - B_{M-1}^{n+\frac{1}{2}}\right) - \Delta t C_M^{n+\frac{1}{2}}$$

There is a definite limit on the magnitude of time step Δt. Any small disturbance that occurs in the basin, such as local change in elevation of the water surface, will travel in both directions as a wave with velocity $\sqrt{gh}$ with respect to the water. If in addition the water in the base basin is moving with velocity u, the wave will travel at velocity $u \pm \sqrt{gh}$ with respect to the basin

In order to represent the physical problem correctly, the time step corresponding to the grid spacing Δx must lie within the zone of influence

This requires that

$$\Delta t \leq \frac{\Delta x}{|u| + \sqrt{gh}}$$

Solution Parameters

- Channel width W = 1.0 m
- Channel length L = 20 m
- Initial water height $h_0 = 12\ m$
- Initial Velocity $V_0 = 0$
- Friction coefficient $C_f = 0.005$
- Wave period $T = 2\ s$
- Wave Amplitude = 1 m
- Gravitational acceleration $g = 9.81\ Nm/s^2$
- Time step $\Delta t = 0.0008\ s$
- Solve for 10 cycles

Results and Discussion



Figure 3: Allowable Time Step at Each Iteration (Time)

In order to obey CFL condition for stability, time step of 0.0008 second is used in calculations. As seen from figure 3, maximum allowable time step on each iteration is slightly larger than 0.0008 seconds. A maximum time step can be calculated and used for each iteration to get a faster convergence. In addition, it is necessary to perform this operation when the time step is unknown, or to be determined.



Figure 4: Height of Water at both ends of Domain

The water height is at the sea-side(reservoir side) is taken a sine function with wave amplitude of 1 m and period of 2 s. It is assumed that the reservoir is so strong that wont be affected from the reflected waves. The result is typical sine function.

On the other hand, from figure 4, The water height at the wall side is strongly affected from incoming and reflected outgoing waves. So that, water amplitude can reach higher and lower levels than the source. Note that there is no disturbance on water level until 2nd second. Which means the information from the reservoir reaches to wall side in 2 seconds. This confirms that the additional term $\sqrt{gh}$ in the CFL condition.



Figure 5: Surface profile at the end of 10th cycle



Figure 6: Surface Profiles with Different Grid Numbers

In order to prove that grid resolution is good enough, some different grid numbers are tried. 1001, 2001, 4001 and 8001 divisions are used with time step of 0.0016, 0.0008, 0.0004, and 0.0002 accordingly to stay in the stability region.

According to figure 6 water surface profiles are more or less the same, but with 1001 divisions there are high fluctuations in surface. And the fluctuations are getting smaller with increased grid numbers. With 8001 divisions and time step of 0.0002 s the fluctuations are completely removed, and solution is acceptable. The slight difference in the surface profiles is caused by these fluctuations, and represents erroneous solutions.

Source Code

```
      program WaterWaves
c..Taha Yaşar Demir /1881978
c..CE-580 - Homework #5
      parameter(mx=20001)
      common/para/ dx,dt,W,Pw(mx),rL,Cf,T,cycles,T_max,Hi,Vi,g,time
      common/flow/ u(mx),h(mx),a(mx),b(mx),c(mx),Pwt(mx),pi,N
      common/midstep/ Utmp(mx),Htmp(mx),Atmp(mx),Btmp(mx),Ctmp(mx)

      open(11,file='cfl.dat')
      open(12,file='h.dat')
      open(13,file='surface.dat')

      call init
      time = 0.
      do while(time.le.T_max)
         call output(1)
         time = time + 0.5*dt
         call mid_continuity
         call mid_momentum
         time = time + 0.5*dt
         call continuity
         call momentum
         print*, time
      end do

      call output(2)

      close(11)
      close(12)
      close(13)

      stop
      end
c------------------------------------------------------------------------
      subroutine init
      parameter(mx=20001)
      common/para/ dx,dt,W,Pw(mx),rL,Cf,T,cycles,T_max,Hi,Vi,g,time
      common/flow/ u(mx),h(mx),a(mx),b(mx),c(mx),Pwt(mx),pi,N
      common/midstep/ Utmp(mx),Htmp(mx),Atmp(mx),Btmp(mx),Ctmp(mx)

      pi    = 22./7.
      T     = 2. ! s
      cycles = 10.
      T_max  = T*cycles ! s
      dt    = 0.0008 ! s Change it according to grid number
      W      = 1. ! m - width of channel
      rL    = 20. ! m - lenght of channel
      Hi    = 12. ! m - initial water heigth
      Vi    = 0. ! m/s -initial velocity
      N     = 2001 ! grid number / Try other Numbers
```

```
          Cf    = 0.005
          g     = 9.81 ! N.m/s^2 gravitational acceleration
          dx    = rL/(N-1)
          do i=1,N
               Pw(i)= 2*h(i) + W ! no need
               u(i) = Vi
               h(i) = Hi
               a(i) = u(i)*h(i)
               b(i) = h(i)*u(i)**2 + 0.5*g*h(i)**2 ! no need
               c(i) = (Cf*u(i)*abs(u(i))*Pw(i))/(2*W) ! no need
          enddo

          return
          end
c-----------------------------------------------------------------------
          subroutine mid_continuity
          parameter(mx=20001)
          common/para/ dx,dt,W,Pw(mx),rL,Cf,T,cycles,T_max,Hi,Vi,g,time
          common/flow/ u(mx),h(mx),a(mx),b(mx),c(mx),Pwt(mx),pi,N
          common/midstep/ Utmp(mx),Htmp(mx),Atmp(mx),Btmp(mx),Ctmp(mx)

     Htmp(N) = Hi + 0.5*sin(2*pi*time/T) ! Resorvoir side is input wave function
          Htmp(1) = h(1) - (0.5*dt/dx)*(a(2)-a(1)) ! forward difference for first node
          do i=2,N-1
               Htmp(i) = 0.5*(h(i+1)+h(i-1)) - (0.25*dt/dx)*(a(i+1)-a(i-1))
c              print*, i,Htmp(i),h(i+1),h(i-1),a(i+1),a(i-1)
          enddo

          return
          end

c-----------------------------------------------------------------------
          subroutine mid_momentum
     parameter(mx=20001)
          common/para/ dx,dt,W,Pw(mx),rL,Cf,T,cycles,T_max,Hi,Vi,g,time
          common/flow/ u(mx),h(mx),a(mx),b(mx),c(mx),Pwt(mx),pi,N
          common/midstep/ Utmp(mx),Htmp(mx),Atmp(mx),Btmp(mx),Ctmp(mx)

          Atmp(1) = 0. ! boundary condition
          Utmp(1) = 0. ! boundary condition
          Btmp(1) = 0.5*g*Htmp(1)**2
          Pwt(1)  = 2*Htmp(1) + W
          Ctmp(1) = 0.

          Atmp(N) = a(N) - (0.5*dt/dx)*(b(N)-b(N-1)) - 0.5*dt*c(N)
          Utmp(N) = Atmp(N) / Htmp(N)
          Btmp(N) = Htmp(N)*Utmp(N)**2 + 0.5*g*Htmp(N)**2
          Pwt(N)  = 2*Htmp(N) + W
          Ctmp(N) = (Cf*Utmp(N)*abs(Utmp(N))*Pwt(N))/(2*W)
c         print*, time,Ctmp(N),Utmp(N),Btmp(N),Htmp(N)
          do i=2,N-1
           Atmp(i) = 0.5*(a(i+1)+a(i-1))-(0.25*dt/dx)*(b(i+1)-b(i-1))
```

```
     +              -0.5*dt*c(i)
                    Utmp(i) = Atmp(i)/Htmp(i)
                    Btmp(i) = Htmp(i)*Utmp(i)**2 + 0.5*g*Htmp(i)**2
                    Pwt(i)  = 2*Htmp(i) + W
                    Ctmp(i) = (Cf*Utmp(i)*abs(Utmp(i))*Pwt(i))/(2*W)
c                   print*, time,i,Atmp(i),Utmp(i),Btmp(i),Ctmp(i)
           enddo

           return
           end
c-------------------------------------------------------------------
           subroutine continuity
           parameter(mx=20001)
           common/para/ dx,dt,W,Pw(mx),rL,Cf,T,cycles,T_max,Hi,Vi,g,time
           common/flow/ u(mx),h(mx),a(mx),b(mx),c(mx),Pwt(mx),pi,N
           common/midstep/ Utmp(mx),Htmp(mx),Atmp(mx),Btmp(mx),Ctmp(mx)


           h(1) = h(1) - (dt/dx)*(Atmp(2)-Atmp(1))
           do i=2,N-1
                    h(i) = h(i) - (0.5*dt/dx)*(Atmp(i+1)-Atmp(i-1))
           enddo
           h(N) = Hi + 0.5*sin(2*pi*time/T)



           return
           end
c-------------------------------------------------------------------
           subroutine momentum
           parameter(mx=20001)
           common/para/ dx,dt,W,Pw(mx),rL,Cf,T,cycles,T_max,Hi,Vi,g,time
           common/flow/ u(mx),h(mx),a(mx),b(mx),c(mx),Pwt(mx),pi,N
           common/midstep/ Utmp(mx),Htmp(mx),Atmp(mx),Btmp(mx),Ctmp(mx)

           a(1)  = 0.
           u(1)  = 0.
           b(1)  = 0.5*g*h(1)**2
           c(1)  = 0.
           Pw(1) = 2*h(1) + W

           a(N) = a(N) - (dt/dx)*(Btmp(N)-Btmp(N-1)) - dt*Ctmp(N)
           u(N) = a(N)/h(N)
           b(N) = h(N)*u(N)**2 + 0.5*g*h(N)**2
           Pw(N)= 2*h(N) + W
           c(N) = (Cf*u(N)*abs(u(N))*Pw(N-1))/(2*W)

c          print*, time,a(N),b(N),u(N),h(N),Btmp(N),Btmp(N-1)

           do i=2,N-1
                    a(i) = a(i) - (0.5*dt/dx)*(Btmp(i+1)-Btmp(i-1)) - dt*Ctmp(i)
c                   print*, Btmp(i+1),Btmp(i-1),Ctmp(i)
```

```
            u(i) = a(i) / h(i)
c           print*, i,time,u(i),a(i),h(i)
            b(i) = h(i)*u(i)**2 + 0.5*g*h(i)**2
            Pw(i)= 2*h(i) + W
       c(i) = (Cf*u(i)*abs(u(i))*Pw(i))/(2*W)
      enddo

      return
      end
c-------------------------------------------------------------------
      subroutine output(m)
      parameter(mx=20001)
      common/para/ dx,dt,W,Pw(mx),rL,Cf,T,cycles,T_max,Hi,Vi,g,time
      common/flow/ u(mx),h(mx),a(mx),b(mx),c(mx),Pwt(mx),pi,N
      common/midstep/ Utmp(mx),Htmp(mx),Atmp(mx),Btmp(mx),Ctmp(mx)
      real cfl(N),x

      do i=1,N
            cfl(i) = dx/(abs(u(i))+sqrt(g*h(i)))
      enddo

      write(11,*) time,minval(cfl)
      write(12,*) time,h(1),h(N)

      x = 0.
      if (m.eq.2) then
            do i=1,N
                  write(13,*) x,h(i)
                  x = x + dx
            enddo
      endif

      return
      end
c-------------------------------------------------------------------
```

Fortran Code Used for Calculations